

More declarative tabling in Prolog using multi-prompt delimited control

Samer Abdallah (samer@jukedeck.com)

Jukedeck Ltd.

August 28, 2017

Abstract

Several Prolog implementations include a facility for *tabling*, an alternative resolution strategy which uses memoisation to avoid redundant duplication of computations. Until relatively recently, tabling has required either low-level support in the underlying Prolog engine, or extensive program transformation (de Guzman et al., 2008). An alternative approach is to augment Prolog with low level support for continuation capturing control operators, particularly *delimited continuations*, which have been investigated in the field of functional programming and found to be capable of supporting a wide variety of computational effects within an otherwise declarative language.

This technical report describes an implementation of tabling in SWI Prolog based on delimited control operators for Prolog recently introduced by Schrijvers et al. (2013). In comparison with a previous implementation of tabling for SWI Prolog using delimited control (Desouter et al., 2015), this approach, based on the functional memoising parser combinators of Johnson (1995), stays closer to the declarative core of Prolog, requires less code, and is able to deliver solutions from systems of tabled predicates incrementally (as opposed to finding all solutions before delivering any to the rest of the program).

A collection of benchmarks shows that a small number of carefully targeted optimisations yields performance within a factor of about 2 of the optimised version of Desouter et al.'s system currently included in SWI Prolog.

1 Introduction and background

Tabling, or memoisation (Michie, 1968) is a well known technique for speeding up computations by saving and reusing the results of earlier subcomputations instead of repeating them. In other contexts, this strategy is called ‘dynamic programming’, and, when used effectively, can reduce algorithms with exponential time complexity to polynomial or linear complexity.

In the field of logic programming, memoisation is referred to as *tabling*, and usually includes the ability to handle recursive, non-deterministic predicates, including so-called *left-recursive* predicates that would otherwise result in non-termination using depth-first search strategies such as SLD resolution, as provided,

for example, in Prolog. Alternative strategies such as OLDT resolution (Tamaki and Sato, 1986) and SLG resolution (Chen and Warren, 1993) have been available for some time in Prologs such as XSB and YAP, but it was recognised well before that the forms of deduction that these methods embody are closely related to the efficient parsing algorithm invented by Earley (1970), the generalisation of which was dubbed ‘Earley deduction’ (Warren, 1975; Pereira and Warren, 1983; Porter, 1986).

In the functional programming community, the problem of efficient parsing has been addressed using memoising parser combinators (Norvig, 1991; Leermakers, 1993). Abdallah (2017) gives an overview of this approach and focusses on the method of Johnson (1995), which is particularly interesting in that it handles left-recursive grammars using essentially the same mechanism as Earley’s algorithm and, by extension, the OLDT and SLG resolution methods used in tabled Prologs. Those methods rely on the *suspension* of chains of deduction, which can then be resumed under different conditions. Johnson’s method (implemented in the functional programming language Scheme) uses continuation passing style (Sussman and Steele, 1975) to achieve the same effect, allowing threads of computation to be suspended and resumed multiple times. Indeed, transformation of a program into continuation passing style, or the use of continuation capturing control operators such as *call-with-current-continuation* in Scheme, is a very powerful approach to implementing many computational effects within a declarative language (Filinski, 1999).

Continuation passing style has been investigated as a basis for implementing Prolog itself (Tarau and Dahl, 1994a; Lindgren, 1994; Neumerkel, 1995) and for providing useful computational effects from within in Prolog (Tarau and Dahl, 1994b, 1998). Continuations were also the basis of de Guzman et al.’s (2008) implementation of tabling, which required a fairly extensive source level program transformation to make the continuations available.

Recently, continuation capturing control operators have been implemented in hProlog (Schrijvers et al., 2013), and subsequently used to implement tabling completely in Prolog, without any further low level support (Desouter et al., 2015). This report describes a similar approach, first translating Johnson’s (1995) functional memoisation into pure, declarative Prolog (in Sec. 3), then (in Sec. 4) making minimal changes to implement tabling as it is more usually seen in Prolog, while retaining most of the declarativeness of the functional version, and finally (in Sec. 5) analysing the computation complexity of the system and making a small number of targeted optimisations to yield acceptable performance. Concluding remarks are to be found in Sec. 6 and supplementary code in the appendices.

The system presented here was developed using SWI Prolog (Wielemaker et al., 2012), which includes an implementation of delimited continuations derived from Schrijvers et al.’s (2013) original implementation for hProlog. We assume some familiarity with Prolog in general. In the program listings that follow, the code has been rendered using typographic features not available in the real source; in particular :- is rendered as \leftarrow , -> as \rightarrow , --> as \longrightarrow , and variables with numeric suffixes are rendered with subscripts. In addition, Prolog lambda expressions (Neumerkel, 2009) are typeset, for example, as $\lambda X.Y.Goal$, but coded as $\backslash X^{\wedge} Y^{\wedge} Goal$.

2 Multi-prompt delimited control in Prolog

Continuation passing style and continuation-based control operators have been well studied in the context of lambda calculus and functional programming (Sussman and Steele, 1975). The idea of delimited continuations was arrived at independently by Felleisen (1988) and Danvy and Filinski (1990). Multiple, first class, independently addressable prompts were introduced by Dyvbig et al. (2005) in their monadic Haskell implementation, which was recast as an OCaml library to be used in ‘direct style’ (i.e. using neither monadic nor explicit continuation passing styles) by Kiselyov (2012). The basic operation of multi-prompt delimited control can be described in terms of the *new_prompt*, *push_prompt* and *shift* operators provided by this library, which have the following OCaml types (in which Greek letters denote type variables):

$$\begin{aligned} \text{new_prompt} & : \text{unit} \rightarrow \alpha \text{ prompt} \\ \text{push_prompt} & : \alpha \text{ prompt} \rightarrow (\text{unit} \rightarrow \alpha) \rightarrow \alpha \\ \text{shift} & : \alpha \text{ prompt} \rightarrow ((\beta \rightarrow \alpha) \rightarrow \alpha) \rightarrow \beta \end{aligned}$$

Given a prompt p with answer type α , created using *new_prompt*, a thunk $f : \text{unit} \rightarrow \alpha$ is evaluated in a context delimited by the prompt p by calling *push_prompt* p f , which plays the role of *reset* in Danvy and Filinski’s (1990) framework. Calls to *push_prompt*, using the same or differing prompts, can be nested arbitrarily. If, inside f , there is a call to *shift* p h , the continuation out to the nearest instance of *push_prompt* p is captured as a function $k : \beta \rightarrow \alpha$, and passed to the handler h , whose return value is then returned from the enclosing *push_prompt*. The captured continuation includes an implicit enclosing *push_prompt* p , which hides any further calls to *shift* p it may contain, and means that, from the outside, it looks like a pure function, which can be called zero, once, or many times. During a call to the handler h , the original outer prompt stays in place, so that any calls to *shift* p made by the handler do not escape the original delimited context. Other frameworks provide a different set of control operators, but they are all closely related and expressible in terms of each other (Shan, 2004).

Schrijvers et al.’s (2013) implementation of delimited continuations for Prolog provides two primitives, *reset/3* and *shift/1*. If a Prolog goal is called as *reset*(*Goal*, *Ball*, *Cont*), with both *Ball* and *Cont* unbound, then *Goal* is executed inside a delimited context. If it completes successfully, then *Cont* is unified with 0. If *shift/1* is called inside *Goal*, then *Cont* is bound to a callable representation of the continuation from that point out to the enclosing *reset*, *Ball* is bound to the argument of *shift/1*, and control is passed to the code following the *reset*. The continuation *Cont* can be called subsequently, resuming the interrupted execution of the original goal, but if that includes further calls to *shift/1*, then it must be called inside a new *reset*, since, unlike in the functional version of *shift*, the captured continuation does *not* re-establish the delimited context.

Calls to *reset/3* can be nested. In this case, a *shift* is handled by the nearest enclosing *reset*, *unless* that *reset* was called with a partially instantiated *Ball* parameter that does not unify with the parameter to *shift/1*. In this case, the *shift* travels outwards until a *reset* with a unifiable *Ball* parameter is found. This mechanism is intended to allow different computational effects to be implemented by nesting calls to *reset/3*, each matching a different set of *Ball* terms and each

accompanied by a suitable effect handler to deal with the requested effect and then resume the computation by calling the continuation.

We will begin with a module *delimcc* that provides an alternative interface to delimited control that uses a result term that can be tested by pattern matching on the head functor, instead of the ‘default’ *Cont* produced by *reset/3*, which requires a test and a cut to be processed correctly. It also and also avoids the non-logical behaviour with partially instantiated *Ball* terms by taking instead an explicit first class prompt as in Dyvbig et al.’s (2005) and Kiselyov’s (2012) libraries.

```

← module(delimcc, [p_shift/2, p_reset/3, pr_shift/2, pr_reset/3]).

← meta_predicate p_reset(+,0,-).
p_reset(Prompt, Goal, Result) ←
  reset(Goal, Prompt-Signal, Cont),
  ( Cont == 0 → Result = done
  ; Result = susp(Signal, Cont)
  ).

p_shift(Prompt, Signal) ← shift(Prompt-Signal).

```

Prompt is a separate argument to both *p_reset/3* and *p_shift/3*, and must be a ground term. The result of the call to *p_reset/3* is either *done* to indicate no continuation was captured, or *susp*(*Signal*, *Cont*) to indicate that *Signal* was sent to this prompt and the captured continuation is *Cont*.

With the aim of drawing a closer parallel between the Prolog and functional implementations of delimited control, the module also contains *pr_reset/3* and *pr_shift/2*, which build on *p_reset/3* and *p_shift/2*. Instead of the *Goal* argument of *p_reset/3*, *pr_reset/3* accepts a callable term *Pred* representing a unary predicate. This is called with a single unbound argument, as in *call*(*Pred*, *X*), and plays the role of the thunk $f : unit \rightarrow \alpha$ taken by *push_prompt* in the OCaml library. Instead of the term *Signal* taken by *p_shift/2*, *pr_shift/2* takes a term *H* representing a binary predicate and is called as *call*(*H*,*K*,*X*), where *K* represents the continuation as a unary predicate and *X* is the unbound variable that will be the final answer returned by *pr_reset/3* in its third argument.

```

← use_module(lambdaki).

← meta_predicate pr_reset(+,1,-).
pr_reset(Prompt, Pred, Result) ←
  p_reset(Prompt, call(Pred, X), Status),
  pr_cont(Status, Prompt, X, Result).

pr_cont(done, _, X, X).
pr_cont(susp(H, Cont), Prompt, X, Result) ←
  pr_reset(Prompt, call(H, pr_reset(Prompt,  $\lambda X$ .Cont)), Result).

← meta_predicate pr_shift(+,2).
pr_shift(Prompt, H) ← p_shift(Prompt, H).

```

Note that, before being passed to the handler H , the raw continuation $Cont$ is wrapped in its own delimited context and expressed as a unary predicate by using a lambda term $\lambda X. Cont$ (see the supporting module *lambdaki* in the appendix A.1), which turns the output variable X into a parameter, and introduces exactly the right copying semantics to allow the continuation to be called multiple times without the interference that could be caused by the possible binding of variables inside the original continuation term $Cont$. Together these have the effect of making the continuation look pure from the outside. Note also that the handler is called inside a delimited context and so may use computational effects too, just as in the functional version of *shift*.

3 Memoisation functional style

3.1 State handling using delimited control

Stateful computations are sometimes handled in Prolog by using two extra arguments to pass the state in and out of all predicates that need to manipulate it. This has the advantage of preserving a declarative reading of the program, without resorting to a procedural reading involving side effects, but done manually, it can become tedious and error prone. The *definite clause grammar*, or DCG syntax Pereira and Warren (1980) recognised by most Prologs makes this much easier by hiding the extra arguments—they are inserted and threaded through all goals automatically when source files containing DCG notation are read. However, using DCG syntax in this way requires that all the code be lifted into DCG form even if many parts do not need to manipulate the state. Delimited continuations allow statefulness to be provided as a computation effect *within* the delimited context, but to appear pure (like a DCG) from outside the context. Schrijvers et al. (2013) use state handling as one of their examples, and we will use a similar approach here, but using the multi-prompt control predicates introduced in the previous section to allow the safe nesting of state handling contexts within each other or other effect handlers. The module *ccstate* provides, within the context of *run_state/3*, *app/2* to manipulate the state (using an arbitrary binary predicate to model a state transition) as a computational effect.

```

← module(ccstate, [run_state/4, app/2]).
← use_module(library(delimcc)).

← meta_predicate app(+,2).
app(Pr,P) ← p_shift(Pr,app(P)).

← meta_predicate run_state(+,0,?,?).
run_state(Pr, Goal, S1, S2) ←
  p_reset(Pr, Goal, Status),
  cont_state(Status, Pr, S1, S2).

cont_state(done, _, S, S).
cont_state(susp(app(P),Cont), Pr, S1, S3) ←
  call(P, S1, S2),
  run_state(Pr, Cont, S2, S3).

```

The predicate *run_state/4* reifies this effect as two extra arguments, that is, it makes stateful effects inside the given goal look like the result of a pure predicate from the outside, with the last two arguments carrying the initial and final states.

This interface provides access to a single ‘blob’ of state, but it is quite straightforward to build on top of it an effect handler than instead provides an unbounded supply of references to mutable cells, which can then be manipulated independently:

```

← use_module(library(data/store)).

← meta_predicate run_ref(0), ref_app(+,2).
run_ref(Goal) ← store_new(S), run_state(ref, Goal, S, _).

ref_new(X,R) ← app(ref, store_add(X,R)).
ref_app(R,P) ← app(ref, store_apply(R,P)).
ref_upd(R,X,Y) ← app(ref, store_upd(R,X,Y)).

```

The module *library(data/store)* (see appendix A.5) is included in the SWI Prolog add-on package *genutils* and provides a pure, declarative implementation of a reference-value store using immutable tree data structures.

3.2 Memoised nondeterminism using lists

The facilities described so far are just enough to implement memoisation as a reasonably direct port of the functional, continuation-based approach implemented monadically in OCaml by Abdallah (2017), which was in turn based on Johnson’s (1995) explicit CPS approach. It supports the memoisation of binary predicates (with one input and one output), which may be recursive or indeed left-recursive. Nondeterminism is invoked using *choose/2*, implemented by the continuation handler *choose/4*, and finally reified as a list of alternative results. (Library *rbutils* is part of the *genutils* add-on package mentioned above, and provides additional tools for working with the red-black balanced binary trees implemented in the SWI Prolog built-in library *rbtrees*. See appendix A.5 for more details.)

```

← module(ccmemo, [run_memo/2, choose/2, memo/2]).

← use_module(library(rbutils)).
4 ← use_module(library(delimcc)).
← use_module(library(ccstate)).
← use_module(lambdaki).

8 ← meta_predicate memo(2,-).
memo(P, ccmemo:mem_call(P,R)) ←
    rb_empty(T),
    ref_new(T,R).
12
mem_call(P,R,X,Y) ← pr_shift(memo, h_mem(P,R,X,Y)).
choose(Xs,X) ← pr_shift(memo, h_choose(Xs,X)).

```

```

← meta_predicate run_memo(1,-).
16 run_memo(P,Result) ← pr_reset(memo, to_list(P), Result).
to_list(P,[X]) ← call(P,X).

h_choose(Xs,X,K,Ys) ← foldl(call_append(λX.K),Xs,[],Ys).
20
h_mem(P,R,X,Y,K,Ans) ←
  K_y = λY.K,
  ref_upd(R, T1, T2),
24 ( rb_upd(X, entry(Ys,Conts), entry(Ys,[K_y|Conts])), T1, T2)
  → rb_fold(fst_call_append(K_y), Ys, [], Ans)
  ; rb_empty(EmptySet),
  rb_add(X, entry(EmptySet,[]), T1, T2),
28 call(P,X,Y1),
  ref_app(R, rb_upd(X, entry(Ys,Conts), entry(Ys2,Conts))),
  ( rb_add(Y1, t, YS, Ys2)
  → foldl(flip_call_append(Y1), [K_y|Conts], [], Ans)
32 ; Ans = [], Ys2 = Ys
  )
  ).

36 fst_call_append(K_y,Y_-,A1,A2) ← call_append(K_y,Y,A1,A2).
flip_call_append(Y,K_y,A1,A2) ← call_append(K_y,Y,A1,A2).
call_append(K_y,Y,A1,A2) ← call(K_y,Y,A), append(A1,A,A2).

```

The context for running memoised nondeterministic computations is provided by `run_memo/2`, which installs a prompt named `memo` and calls the supplied predicate `P` using wrapper `to_list/2`, which returns the result of `P` in a (singleton) list. Nondeterminism is handled by `h_choose/4`, which applies the continuation `K` to each item in the supplied list of alternatives `Xs`, concatenating the lists of results produced by each application of the continuation.

A memoised version of a binary predicate is prepared using `memo/2`, which creates a new mutable reference containing the initially empty memo table for that predicate. The memoised version is a call to `mem_call/4`, which, invokes the continuation handler `h_mem/6`. This works very much like Abdallah’s (2017) functional version: if the memoised predicate has already been applied to the input `X`, then there will already be an entry in the memo table `T1` (line 25), in which case the newly captured continuation, in the form of a two-argument lambda term `K_y`, is added to the list of ‘consumer’ continuations associated with `X`, and then applied to all the results produced by the base predicate `P` so far (line 26), collecting all the final results in the list `Ans`. If this is the first time the memoised predicate has been applied to `X`, then a new ‘producer’ is initiated by adding an entry to the memo table (line 28), calling the base predicate `P`, and, for each result `Y1`, producing an empty answer list if `Y1` is already in the results table (line 33) or sending `Y1` to all of the continuations waiting for the result of applying `P` to `X`. The whole of the handler predicate `h_mem/6` is called inside a `memo` prompt, so that any nondeterminism or memoisation effects triggered inside the call to `P` (line 29) result in the following lines of code being executed multiple times and the answers being collected in the correct way.

As an example of how this framework can be used, the following illustrates a left-recursive predicate *path/2* which implements the transitive closure of *edge/2*.

```

← use_module(ccmemo).

path_memo(Start, Ends) ←
4   memo(path(Path), Path),
   run_memo(call(Path,Start), Ends).

path(Path, N1, N3) ←
8   choose([0,1 ], U),
   ( U = 0 → edge(N1,N3)
   ; U = 1 → call(Path,N1,N2), edge(N2,N3)
   ).
12
edge(a,X) ← choose([b,c ], X).
edge(b,d).
edge(c,d).
16 edge(d,X) ← choose([], X).

```

The predicate *path_memo/2* must be called inside a context providing mutable references, e.g., as *run_ref(path_memo(Start, Ends))*. A few observations on this code are in order. Firstly, *path/3* is written in *open recursive* style; that is, rather than call itself directly when recursion is required, it calls the first argument *Path*, which is assumed to represent the memoised recursive path predicate. The recursion is tied-up without using an explicit fixed-point operator simply by memoising *path(Path)* on line 4 and unifying the result with *Path* (results in a cyclic term). Secondly, all nondeterminism in the program, both in *edge/2* predicate and *path/3*, must be represented using *choose/2*, rather than using Prolog’s built-in nondeterminism. This results in a rather non-idiomatic Prolog style, for example, we cannot simply omit the clause for *edge(d,X)* on line 16 and rely on normal Prolog failure to indicate that there is no edge from *d*; we must use *choose([],_)* to express this fact. Thirdly, although not a problem in this example, the restriction to memoising only binary ‘input-output’ predicates does not fit well with the Prolog norm of using arbitrary arity predicates without restrictions on which arguments may be considered ‘inputs’ or ‘outputs’. In short, we have not really implemented *Prolog* style tabling, but rather transplanted a functional idiom into Prolog. These deficiencies are remedied in the next section.

4 Idiomatic Prolog tabling

4.1 Tabling with non-backtrackable state

The Prolog implementation of delimited continuations interacts unproblematically with Prolog’s nondeterminism and backtracking: any choice points created inside a *reset/3* are preserved on leaving the delimited context, and subsequent failure will cause backtracking to a choice point, undoing of variable bindings, and re-execution of subsequent goals both inside and outside the delimited context. This means, for example, the state handling provided by *run_state/4* interacts

with nondeterminism just as a DCG or other pure Prolog predicate would, with state transitions being undone on backtracking.

Tabling relies on sharing state *across* alternative branches of computation. The implementation in the previous section dealt with this by eschewing Prolog nondeterminism entirely and using lists to accumulate alternative results. If we wish to use Prolog’s nondeterminism to represent these alternatives, then the state of the tables must be managed in such a way that changes are not reversed on backtracking.

These two ways of combining of state with nondeterminism (usually referred to as *backtrackable* and *non-backtrackable* state respectively) can be represented in a functional setting using *monad transformers* (Liang et al., 1995). Within that framework, the two computational effects can be layered in either order, with the two resulting monadic types making clear the difference between them. If σ is the type of the state and nondeterminism is represented as a list, the type of a monadic computation yielding values of type α is $\sigma \rightarrow (\alpha \times \sigma)$ *list* when layering state on top of nondeterminism, but $\sigma \rightarrow \alpha$ *list* $\times \sigma$ with nondeterminism on top of state. The former corresponds to backtrackable state, since there is a state associated with each nondeterministic alternative, while the latter corresponds to non-backtrackable state, since there is a single final state resulting from the traversal process that produced the list of alternatives. The fact that the final state depends on the traversal process does not compromise the purity of the computation, because the mapping from initial state to results plus final state is still deterministic and free of side effects.

Using Prolog’s nondeterminism to represent a collection of results, that is, using a predicate that succeeds once for each result, the requirements for declarativeness are more stringent: the model theoretic meaning of a predicate is a *set* of tuples, invariant with respect to the strategy used to find them. In general, the use of non-backtrackable state means that the sequence of state transitions depends on the order in which the proof tree of a goal is explored, so it follows that, if a pure logical reading is to be maintained, the state should not be exposed in the interface of a predicate that uses it internally.

SWI Prolog provides a number of ways to manage non-backtrackable state, including dynamic predicates (*assert/1* and *retract/1*), a ‘recorded database’, and non-backtrackable global variables (*nb_setval/2* and *nb_getval/2*). All three of these (necessarily) involve a departure from pure Prolog semantics. Using *run_nb_state/4* (implementation given in appendix A.2), the scope of this impurity and the associated dangers of unprotected global mutable state are limited to the introduction of one mutable cell accessed via a carefully controlled computational effect, with essentially the same interface as the backtrackable state effect provided by *run_state/4* in module *ccstate* (Sec. 3.1).

4.2 Handling arbitrary predicate arity and mode

The next requirement for an idiomatic Prolog interface to tabling is the ability to table predicates of any arity and with any pattern of argument instantiation (usually referred to as *mode*). This can be done by maintaining a separate table for each distinct combination of input terms and output variables encountered in calls to the predicate, that is, treating calls which are *variants* (Sterling and Shapiro, 1994) of each other as forming an equivalence class. For example, *foo(a, X, Y)* and *foo(a, U, V)* are variants of each other, but *foo(a, X, Y)*, *foo(b, X, Y)* and *foo(a, X, X)*

are not. Variant calls will have the same set of solutions and can share the same table. In the sequel, an equivalence class of calls which are all variants of each other will be called a ‘variant class’.¹ The safest way to manipulate variant classes without resorting to extra-logical operations is to represent them as ground terms as soon as possible, which can be done using *numbervars/3*. This replaces each distinct variable with a term ‘ $\$VAR'(N)$ ’, where N is an integer. Thus, in the examples above $foo(a, X, Y)$ will be represented as $foo(a, \$VAR'(0), \$VAR'(1))$, while $foo(a, X, X)$ will be represented as $foo(a, \$VAR'(0), \$VAR'(0))$.

Ground terms representing variant classes can be used as keys to their associated tables in an associative map which is in turn stored in the non-backtrackable mutable cell provided *run_nb_state/4*, described in Sec. 4.1. Because the key is derived from the predicate call, there is no need for an explicit allocation step as there was in the functional version. Note also that the stratification in table access between the binary predicate at the first level and the input argument at the second is gone: each variant class combines predicate and ‘input’ arguments, and is a separate entity at the top level.

The ‘output’ from calling a particular variant class consists of the bindings of the variables in the call, which may be distributed in any way across the parameters of the predicate, including possible partial instantiations of arguments. These variables can be extracted using the standard Prolog extra-logical operation *term_variables/2*. Each solution of a tabled variant class can then be represented as a list of the resulting values of these variables.

4.3 Using Prolog nondeterminism

The final requirement is to use Prolog’s nondeterminism both at the level of the predicates being tabled (instead of the awkward *choose/2* effect in the functional version) and to represent nondeterminism outside the tabled environment (instead of reifying nondeterminism as a list of alternatives). This will have the additional effect that the solutions of tabled predicates can be explored incrementally, instead of being produced all at once. In the module *cctable* below, this behaviour is effected in two places. Firstly, if a call is made that already has a table entry (i.e. a consumer call), instead of calling the consumer continuation for each solution already found and aggregating the final answers, line 27 extracts a solution from the solution set *nondeterministically*, leaving a choice point, and then calls the consumer continuation. Any subsequent failure will result in backtracking to this choice point and application of the continuation to the next solution in the set. Secondly, in *producer/4* (line 37), *call(Generate, Y)* will be nondeterministic, and each time it succeeds, previously seen solutions can be ignored simply by failing in the attempt to add the value to the solution set (line 44), while previously unseen solutions are sent to the registered continuations (including the producer’s continuation) by calling them *disjunctively*, rather than conjunctively (line 40). It is significant that the producer’s continuation K_p is called *first*: this means that from the outside, the tabled predicate produces answers lazily, before all solutions have been found.

¹It should be noted that manipulating terms on the basis of what variables they contain involves stepping up out of the pure, declarative fragment of Prolog in to pure Prolog’s *metalanguage*. The fact that Prolog serves as its own metalanguage means that this distinction is easily overlooked.

```

← module(cctable, [run_tabled/1, cctabled/1]).

← use_module(library(delimcc), [p_reset/3, p_shift/2]).
4 ← use_module(library(ccnbstate), [run_nb_state/4, app/4]).
← use_module(library(rbutils)).
← use_module(lambdaki).

8 ← meta_predicate cctabled(0).
cctabled(Work) ← p_shift(tab, Work).

← meta_predicate run_tabled(0).
12 run_tabled(Goal) ←
    term_variables(Goal, Ans),
    rb_empty(Empty),
    run_nb_state(state, run_tab(Goal, Ans), Empty, _).
16
run_tab(Goal, Ans) ←
    p_reset(tab, Goal, Status),
    cont_tab(Status, Ans).
20
cont_tab(done, _).
cont_tab(susp(Work, Cont), Ans) ←
    term_variables(Work, Y),
24    K =  $\lambda Y.$ Ans.Cont,
    numbervars_copy(Work, VC),
    app(state, new_cont(VC, K, A)),
    ( A = solns(Ys) → rb_in(Y, _, Ys), run_tab(Cont, Ans)
28    ; A = producer → run_tab(producer(VC,  $\lambda Y.$ Work, K, Ans), Ans)
    ).

new_cont(VC, K, solns(Ys), S1, S2) ←
32    rb_upd(VC, tab(Ys,Ks), tab(Ys,[K|Ks]), S1, S2).
new_cont(VC, K, producer, S1, S2) ←
    rb_empty(Ys),
    rb_add(VC, tab(Ys,[K]), S1, S2).
36
producer(VC, Generate, Kp, Ans) ←
    call(Generate, Y),
    app(state, new_soln(VC, Y, Ks)),
40    member(K,[Kp|Ks]), call(K,Y,Ans).

new_soln(VC, Y, Ks, S1, S2) ←
    rb_upd(VC, tab(Ys1, Ks), tab(Ys2, Ks), S1, S2),
44    rb_add(Y, t, Ys1, Ys2).

numbervars_copy(Work, VC) ←
    copy_term_nat(Work, VC),
48    numbervars(VC, 0, _).

```

A few other comments on this code are in order. The state of the system is represented as an associative map (using red-black trees) of variant classes to table entries, each of which contains a set of solutions (also represented as a red-black tree) and a list of consumer continuations. Each instance of the state is an *immutable* data structure: only the cell managed by *run_nb_state/4* is mutable, and its contents are replaced completely on each state transition. These transitions are managed using the predicates *new_cont/5* (to decide what to do with the newly captured continuation on each call to a tabled predicate) and *new_soln/4* (to decide what to do with a new solution from the ‘worker’ goal *Work*). These two predicates are themselves pure, so that the effectful component of the state transition is completely encapsulated by the *app/2* operator.

The variable *Ans* is used to represent the list of ‘outputs’ from the top-level goal initially passed to *run_tabled/1*; this list is extracted from the goal using *term_variables/2*. Similarly, *Y* represents the ‘outputs’ of a given tabled variant class as a list of variables, which are eventually bound when solutions are found.

The continuation *Cont* captured when calling a tabled predicate is a callable term, possibly containing variables shared with the worker goal, representing ‘inputs’ which are expected to be provided by the tabled predicate, and ‘outputs’ which will be bound when the continuation is called. This continuation is stored in the table entry for a variant class in the form of a lambda term $\lambda Y.Ans.Cont$ (line 24) so that it can be called as a binary predicate and called multiple times without binding the variables in the lambda term. Similarly, the worker goal is passed to *producer/4* in the form of a lambda term $\lambda Y.Work$ (line 28), so that it can be called without binding any of these variables.

4.4 Program transformation

While the module *cctable* can be used as-is, by calling tabled goals using the meta-predicate *cctabled/1*, the usual approach to tabling in Prolog is to support a declaration that a certain predicate should be tabled, such that it can then be called without any further decoration. This can be achieved using a very shallow program transformation, similar to that described by Desouter et al. (2015). The implementation is given in the appendix, but result is that the example given in Sec. 3.2 can be written (now using DCG notation for *path/2*) as:

```

← use_module(cctable).
← use_module(ccmacros).

← table path/2.
path → edge; path, edge.

edge(a,b). edge(a,c). edge(b,d). edge(c,d).

```

The *path/2* predicate is transformed via macro expansion on loading to

```

path(N1, N2) ← cctabled('path#'(N1, N2)).
'path#'(N1, N3) ← edge(N1, N3); path(N1, N2), edge(N2, N3).

```

As well as the DCG translation, the original predicate is renamed by adding ‘#’ to the name, while *path/2* is redirected to a tabled call of the renamed predicate.

5 Computational complexity and performance

The system presented above achieves the aim of supporting idiomatic Prolog tabling while consisting almost entirely of pure, declarative Prolog. Unfortunately, its performance is terrible. This is because maintaining non-backtrackable state in Prolog requires a certain amount of term copying: not only is this relatively expensive in itself, it also tends to create more work for the garbage collector. Because the state is held in a *single* mutable variable, the *entire* state is copied every on every transition, even if only a small component of the state has been changed. This creates a tension between the desire for logical purity, which drives us to minimise the number of points of mutation, and the desire for efficiency, which drives us to *factorise* the state into several pieces, so that one component of the state can be modified without incurring the cost of copying the rest. At this point, development becomes a question of pragmatic balance, attempting to achieve acceptable performance with a minimal departure from logical purity, which is, perhaps, a little unsatisfying from a theoretical point of view, but seems to be unavoidable within the framework of current Prolog systems.

5.1 Complexity analysis

We can begin to quantify the costs of various copying operations as follows. Given an application involving tabled predicates, let N_p be the number of ‘producer’ calls; that is, the number of distinct variant classes in use. Let R_c be the average number of ‘consumer’ calls per producer; that is, the number of subsequent calls with a given variant class after the initial one. Finally, let R_s be the average number of solutions per producer. It is clear that the tabling system must call each continuation for each solution for each variant class, so that a minimal complexity for a computation that processes every solution of every tabled call is $O(N_p R_c R_s)$. However, decisions about to represent the state of the system can incur additional copying costs, some of which are quadratic in one or more of these metrics. For example, if a piece of state contains one item for each producer, then copying this state to add a new producer results in an $O(N_p^2 K)$ cost overall, where K is the size of the data structure associated with each producer, which may be $O(1)$, $O(R_c)$, $O(R_s)$ or $O(R_c R_s)$ depending on the representation chosen. The following table summarises this and other costs that may apply:

<i>operation</i>	<i>cost</i>
1. copying each continuation before calling	$N_p R_c R_s$
2. copying each worker to get variant class	$N_p R_c$
3. copying all producers to add a new producer	$N_p^2 K$
4. copying all solutions to add a consumer	$N_p R_c R_s$
5. copying all continuations to add a solution	$N_p R_c R_s$
6. copying all continuations to add a consumer	$N_p R_c^2$
7. copying all solutions to add a solution	$N_p R_s^2$

Of these, the first cannot be avoided, since a single continuation may need to be called multiple times with different ‘inputs’ in the course of a computation. The others result from representational choices; for example, (3) applies when a single term contains information about all producers, which is the case in the implementation of Sec. 4, with $K = R_c R_s$; (6) applies when all the continuations for a particular producer are stored in a single data structure, and similarly

(7) when all the solutions from a producer are stored in one term. Indeed, the implementation of Sec. 4 suffers from all these costs, in particular the quadratic ones, which explains why it is so slow on all but the smallest problems.

There are many strategies for reducing or eliminating these copying costs, several of which are explored in the code repository accompanying this report.² Although a complete description of these is beyond the scope of this report, a few strategies will be discussed below.

Firstly, it is clear from the table of copying costs that for asymptotic performance of $O(N_p R_c R_s)$ it is most important to avoid costs that are quadratic in N_p , R_c or R_s , as these will eventually dominate in large problems. The implementation `cctable(db)` does this by using dynamic predicates to achieve a complete factorisation of the state of the system. Although the overheads associated with using dynamic predicates are larger than those for non-backtrackable global variables, the asymptotic complexity means that performance is still better than some of the implementations (not shown in the table) using non-backtrackable global variables to carry less completely factorised state.

Another way to avoid a quadratic cost when adding solutions is to use the SWI Prolog `tries` library, which allows a single immutable reference to point to a mutable associative map. This was introduced to support the tabling library currently included in SWI Prolog, but is sufficiently general to be used in the implementation `cctable(trie)`, for both the map of variant classes to table entries and the set of solutions for each variant class. Tries are not a suitable data structure for the collection of continuations associated with each producer, as it is not appropriate to treat them as a set or test them for equality. Hence, to avoid quadratic costs here, in `cctable(trie)`, a system of references to growable lists is introduced, built using the low-level `nb_linkval/2` operation to avoid copying the continuations already in the list, in combination with `ccnbref`, a module providing a delimited context for safely managing non-backtrackable mutable storage cells (see appendix A.3).

While developing approaches to managing the collection of continuations efficiently, it became clear that size of the continuation terms can be a significant factor, most obviously in the Fibonacci benchmark, where the total size of the continuations was found to grow quadratically with the size of the problem. Discussions with Jan Wielemaker, the main author of SWI Prolog, lead to the conclusion that there were two causes of this. Firstly (and most easily dealt with) the inclusion of the producer’s continuation (K_p) as a parameter to `producer/4` meant that continuations were being included as values in the stack frames as part of other subsequently captured continuations. This can be avoided by including the producer’s continuation in the data structure used to store the consumers’ continuations, rather than passing it as a parameter. As long as producer’s continuation is kept at the head of the list, the ‘incrementality’ of the system (solutions being delivered as they are found) is preserved.

The second cause was that the process of repeatedly capturing and reactivating continuations along a single conjunctive computational path resulted in continuations which gradually accumulated non-operative stack frames. This lead Jan Wielemaker to introduce optimisations in the low level implementation of `shift/1`, similar in spirit to `tail call` optimisations, to remove these stack frames. This solved the continuation growth problem completely, resulting in

²<https://github.com/samer--/cctable>

approximately linear time complexity on the Fibonacci benchmark using the implementation `cctable(trie)/kp`. An equivalent modification was also applied to the dynamic database version to get the implementation `cctable(db)/kp`.

5.2 An improved implementation

The implementation `cctable(trie)/kp` is given below. Continuations are represented using the $k/3$ functor instead of lambda terms, since a bulk copy (in `lref_get/2`) is faster than copying them individually when calling lambda terms.

```

4 ← use_module(library(delimcc), [p_reset/3, p_shift/2]).
← use_module(ccnbref, [run_nb_ref/1, nbref_new/2]).
← use_module(lambdaki).

8 ← meta_predicate cctabled(0).
cctabled(Head) ← p_shift(tab, Head).

12 ← meta_predicate run_tabled(0).
run_tabled(Goal) ←
    term_variables(Goal, Ans), trie_new(Trie),
    run_nb_ref(run_tab(Goal, Trie, Ans)),

16 run_tab(Goal, Trie, Ans) ←
    p_reset(tab, Goal, Status),
    cont_tab(Status, Trie, Ans).

20 cont_tab(done, _, _, _).
cont_tab(susp(Head, Cont), Trie, Ans) ←
    term_variables(Head, Y), K = k(Y, Ans, Cont),
    ( trie_lookup(Trie, Head, tab(Solns, Conts))
    → lref_add(Conts, K),
      trie_gen(Solns, Y, _),
      run_tab(Cont, Trie, Ans)
    ; lref_new(K, Conts), trie_new(Solns),
      trie_insert(Trie, Head, tab(Solns, Conts)),
      run_tab(producer(λY.Head, Conts, Solns, Ans), Trie, Ans)
    ).

28 producer(Generate, Conts, Solns, Ans) ←
    call(Generate, Y),
    trie_insert(Solns, Y, t),
    lref_get(Conts, Ks),
    member(k(Y, Ans, Cont), Ks), call(Cont).

36 lref_new(K0, Ref) ← nbref_new([K0], Ref).
lref_get(Ref, Xs) ← nb_getval(Ref, Ys), copy_term(Ys, Xs).
lref_add(Ref, K) ←
    nb_getval(Ref, [K0 | Ks]), duplicate_term(K, K1),
    nb_linkval(Ref, [K0, K1 | Ks]).

```

5.3 Benchmarks

Several of the implementation variations described above were tested on a selection of the benchmarks that were used by Desouter et al. (2015) and compared with alternative tabling implementations. Desouter et al.’s system was tested in two forms: the original all-Prolog library, referred to below as *desouter(pl)*, is a port of the original from hProlog to SWI Prolog³, while *desouter(plc)* is an optimised version, with some components re-implemented in C for greater speed. In addition, the tabling implementations in two other Prologs, YAP and B-Prolog were also tested, though it should be noted that these are both much faster Prologs in general and that in both tabling is implemented at a low level. Also, YAP does not support arbitrary precision arithmetic, and so cannot run the Fibonacci benchmark with the parameters indicated. The benchmarks themselves were derived from code available online.⁴

Each of these benchmarks results in a certain pattern of tabled calls, which can be characterised as in Sec. 5.1 in terms of the number of producers, consumers and solutions, as follows:

	N_p	N_c	N_s	R_c	R_s
<i>fib</i> (1000)	1001	998	1001	1.0	1.0
<i>fib</i> (2000)	2001	1998	2001	1.0	1.0
<i>nrev</i> (500)	501	0	501	0.0	1.0
<i>nrev</i> (1000)	1001	0	1001	0.0	1.0
<i>shuttle</i> (2000)	1	2	4001	2.0	4001.0
<i>shuttle</i> (5000)	1	2	10001	2.0	10001.0
<i>ping_pong</i> (10000)	2	2	20002	1.0	10001.0
<i>path_dfst</i> (50)	50	2402	2401	48.0	48.0
<i>path_dfst</i> (100)	100	9802	9801	98.0	98.0
<i>path_dfst_loop</i> (50)	50	4803	4802	96.1	96.0
<i>recognise</i> (20000)	2	2	20000	1.0	10000.0
<i>pyramid</i> (500)	500	995	186751	2.0	373.5
<i>test_joins</i>	1	4	371293	4.0	371293.0
<i>monoidal</i>	35	42195	2664	1205.6	76.1

Notably, the naïve reverse benchmark *nrev*(_) does not involve any consumer calls: each variant class is called just once, producing one answer, and so this benchmark does not benefit from tabling at all. Thus, the timing results give an indication of the minimum overhead introduced by the tabling system. The results are shown in Tab. 1, and a subset of the results illustrated graphically in Fig. 1. The benchmarks were run on a 2012 MacBook Pro with a 2.5 GHz Core i5 processor and 8 GB of memory.

Overall, YAP is the fastest system, with efficient an implementation of SLG resolution as well as a fast underlying Prolog.⁵

³Available at https://github.com/JanWielemaker/tabling_library

⁴https://github.com/JanWielemaker/tabling_benchmarks

⁵Desouter et al. (2015) reported some of the execution times for YAP at 0 ms. The code for running these benchmarks measured the time to produce the *first* solution produced. YAP’s tabling algorithm, like the one presented here, is able to produce solutions as they are found, whereas Desouter et al.’s algorithm (both implementations) finds all solutions before producing the first one. Hence the comparison is a bit misleading, and so to produce the benchmarks here, the code was modified to measure the time to find all solutions.

	<i>yap</i>	<i>bprolog</i>	<i>desouter(plc)</i>	<i>desouter(pl)</i>	<i>cctable(db)</i>	<i>cctable(db)/kp</i>	<i>cctable(trie)</i>	<i>cctable(trie)/kp</i>
<i>monoidal</i>	<u>1.3 s</u>	881	2.2 s	31 s	146 s	133 s	3.8 s	3.9 s
<i>test_joins</i>	674	1.6 s	<u>1.2 s</u>	12 s	⊥	⊥	2.2 s	2.7 s
<i>fib</i> (1000)	–	2	<u>10</u>	110	1.1 s	110	240	20
<i>fib</i> (2000)	–	8	<u>30</u>	230	4.6 s	380	940	40
<i>nrev</i> (500)	17	64	120	1.5 s	<u>60</u>	100	80	90
<i>nrev</i> (1000)	65	253	490	5.5 s	<u>250</u>	380	310	330
<i>path_dfst</i> (50)	3	<u>5</u>	30	500	2.0 s	2.0 s	70	50
<i>path_dfst</i> (100)	21	<u>39</u>	260	3.9 s	65 s	84 s	390	370
<i>path_dfst_loop</i> (50)	15	<u>32</u>	190	2.6 s	23 s	25 s	590	330
<i>ping_pong</i> (10000)	6	3.5 s	<u>40</u>	850	7.5 s	7.9 s	60	60
<i>pyramid</i> (500)	<u>53</u>	38	250	6.5 s	⊥	⊥	380	570
<i>recognise</i> (20000)	15	66 s	<u>70</u>	920	14 s	14 s	170	120
<i>shuttle</i> (5000)	6	1.4 s	<u>30</u>	380	2.3 s	2.2 s	80	80
<i>shuttle</i> (10000)	10	6.0 s	<u>50</u>	790	9.7 s	9.9 s	190	170

Table 1: Execution times for a variety of benchmarks using several tabling implementations. Times are milliseconds unless explicitly stated in seconds. In each row, the best time is typeset in boldface and the second best time is underlined. YAP does not include support for arbitrary precision integer arithmetic and so was not able to run the Fibonacci benchmark at the given sizes. A ⊥ indicates that the benchmark did not complete within the allowed time of 240 s.

B-Prolog uses a different tabling algorithm, *linear tabling* (Shen et al., 2001), with quite different characteristics. It performs very well on many benchmarks but relatively poorly on those which involve a lot of non-determinism.

Of the SWI Prolog-based implementations, the optimised version of Desouter et al.’s library currently included as standard in SWI is the fastest. Compared with this, The implementation *cctable(trie)/kp* developed here is quite competitive, running about half as fast in most benchmarks. Although it shares the optimise trie implementation, it does not benefit from the optimised ‘worklist’ data structure, which stores and reactivates continuations more efficiently than is possible from Prolog code.

It is clear from the results that the steps taken to avoid continuation growth (the *_/kp* versions) are effective, especially on the Fibonacci benchmark, while having a small negative impact on those benchmarks that did not suffer from excessive continuation growth, with *pyramid*(500) suffering the most. It is also clear from the *nrev*(*_*) results that the *cctable(db)* versions, using dynamic predicates, have the smallest overhead when tabling turns out not be useful. However, these two implementations are not competitive on several of the other benchmarks.




Figure 1: Performance of several tabling implementations on a subset of benchmarks. The horizontal axis is the reciprocal of the execution time in seconds, and so longer bars represent faster performance. The timings for YAP have been excluded as they are in almost all cases the fastest.

6 Discussion and conclusions

The implementations presented in this report were largely derived by analogy from the OCaml version described by Abdallah (2017), but as they are built on the same set of delimited control operators as the tabling library of Desouter et al. (2015), they all share many common features: non-backtrackable mutable data structures for mapping variant calls to table data; structures for collecting solutions; structures for collecting continuations and reactivating them when solutions become available, and so on.

However, there are several significant differences. The control structure and scheduling mechanism is simpler and more declarative, with less reliance on side effects. There is much less code: Desouter et al quote a figure of 244 lines of code, excluding the implementation of tries, whereas the main tabling code here is about 40 lines of code, plus about 20 for program transformation and 20 or so in

general purpose libraries. In use, it delivers solutions incrementally, rather than having to collect them all before delivering the first one, which could amount to a significant saving in execution time in certain applications.

It was found while profiling the initial and most declarative implementation that the largest contributions to execution time came from term copying and garbage collection. This is an unavoidable factor when using non-backtrackable state in Prolog, and tends to force the introduction of more separate pieces of mutable state in order to get acceptable performance. However, it is still possible to use delimited control operators, with a relatively low cost, to manage these pieces of mutable state with a minimal impact on the declarativeness of the rest of the program.

The fastest implementation presented here, *cctable(trie)/kp*, is within a factor of about 2 of the tabling system currently included in SWI Prolog, and so with a little more optimisation, could become a viable alternative, bringing the benefits of incrementality, and greater code simplicity and declarativeness.

Acknowledgements

The author would like to thank Jan Wielemaker, the creator of SWI Prolog, for being very responsive in making enhancements to and correcting problems in the implementation of delimited continuations in SWI Prolog, as well as being very helpful in explaining some of the internal details of the implementation.

A Supporting code

Modules *lambdaki*, *ccstate*, *delimcc* and *rbutils* can be found in the SWI Prolog add-on package *genutils*, which can be installed in SWI Prolog with the command *pack_install(genutils)*. The rest of the code presented in this report can be found (in some cases in a slightly expanded and generalised form) at <https://github.com/samer--/cctable>.

A.1 Lambda terms

This module is a minimal implementation of lambda terms, which are callable terms representing predicates taking one or more arguments, and can be useful for high-order programming, reducing the need to define trivial auxiliary predicates. It is a simplified version of Neumerkel’s (2009) *lambda* library, supporting only unary and binary application with no free variables.

```
← module(lambdaki, [(λ)/2, (λ)/3, (·)/3, (·)/4]).
```

```
← meta_predicate λ(1,?), λ(2,?,?).
```

```
← meta_predicate .(?,0,?), . (?,1,?,?).
```

```
λ(M:Hats, A1) ← copy_term(Hats, Copy), call(M:Copy, A1).
```

```
λ(M:Hats, A1, A2) ← copy_term(Hats, Copy), call(M:Copy, A1, A2).
```

```
·(A1, P, A1) ← call(P).
```

```
·(A1, P, A1, A2) ← call(P, A2).
```

A.2 Non-backtrackable mutable state

The module *ccnbstate* implements a delimited context *run_nb_state/4* providing one piece of non-backtrackable mutable state, which can be modified with any binary predicate using *app/2* or retrieved using *get/2*. The global variable used to store the state is protected from accidental interference and is destroyed when the stateful computation is finished.

```
← module(ccnbstate, [run_nb_state/4, app/2, get/2]).
← use_module(library(delimcc)).

← meta_predicate app(+,2).
app(Pr,P) ← p_shift(Pr, app(P)).
get(Pr,P) ← p_shift(Pr, get(P)).

← meta_predicate run_nb_state(+,0,+,−).
run_nb_state(Pr, Goal, S1, S2) ←
  gensym(nbs,K),
  setup_call_cleanup( nb_setval(K, S1),
                    (run(Pr, Goal, K), nb_getval(K, S2)),
                    nb_delete(K)).

run(Pr, Goal, K) ← p_reset(Pr, Goal, Status), cont(Status, Pr, K).

cont(susp(R,Cont), Pr, K) ← handle(R,K), run(Pr, Cont, K).
cont(done, _, _).

handle(get(S),K) ← nb_getval(K,S).
handle(app(P),K) ← nb_getval(K,S1), call(P,S1,S2), nb_setval(K,S2).
```

A.3 Non-backtrackable mutable references

This module implements a delimited context *run_nb_ref/1* for supplying unique references to non-backtrackable mutable storage cells using *nbref_new/2*. These cells are guaranteed to be released when the context is closed.

```
← module(ccnbref, [run_nb_ref/1, nbref_new/2]).
← use_module(library(delimcc), [p_reset/3, p_shift/2]).

nbref_new(Val, Ref) ← p_shift(nbref, new(Val,Ref)).

← meta_predicate run_nb_ref(0).
run_nb_ref(Goal) ←
  setup_call_cleanup(setup(E), run(Goal, E), cleanup(E)).

setup(E) ← gensym(nbref,ID), atom_concat(ID,':',E), nb_setval(E, 0).
delete(E,I) ← atomic_concat(E,I,Ref), nb_delete(Ref).
cleanup(E) ← nb_getval(E, N), nb_delete(E),
  forall(between(1,N,I), delete(E,I)).
```

```
run(Goal, E) ← p_reset(nbref, Goal, Status), cont(Status, E).
```

```
cont(done, _).
```

```
cont(susp(new(Val,Ref),Cont), E) ←  
  nb_getval(E, I), J is I+1, atomic_concat(E, J, Ref),  
  nb_setval(Ref, Val), nb_setval(E, J).  
run(Cont, E).
```

A.4 Program transformation

The module *ccmacros* implements a shallow program transformation to support tabling. Predicates declared ‘tabled’ are renamed (using the special SWI Prolog hook *prolog:rename_predicate/2* to appending a ‘#’ to their given name) and the original predicate name defined as a metacall of the renamed predicate via *cctable/1*, which is assumed to be available in the module where the tabled predicate is defined.

```
← module(ccmacros, [op(1150, fx, table)]).  
← op(1150, fx, table).
```

```
system:term_expansion((← table(Specs)), Clauses) ←  
  foldl_clist(expand_cctab, Specs, Clauses, []).
```

```
foldl_clist(P, (A,B)) → !, call(P,A), foldl_clist(P,B).  
foldl_clist(P,A) → call(P,A).
```

```
prolog:rename_predicate(M:Head, M:Worker) ←  
  ‘$flushed_predicate’(M:‘$cctabled’(_)),  
  call(M:‘$cctabled’(Head)), !,  
  head_worker(Head, Worker).
```

```
expand_cctab(Name//Arity) → !,  
  {A2 is Arity+2},  
  expand_cctab(Name/A2).
```

```
expand_cctab(Name/Arity) →  
  {functor(Head, Name, Arity), head_worker(Head, Worker)},  
  [ (← discontinuous(‘$cctabled’/1))  
  , ‘$cctabled’(Head)  
  , (Head ← cctabled(Worker))  
  ].
```

```
head_worker(Head, Worker) ←  
  Head =.. [H|As], atom_concat(H, ‘#’, W),  
  Worker =.. [W|As].
```

A.5 Description of library predicates

The module *rbutils* is an interface to the predicates in the standard SWI Prolog module *library* (*rbtrees*) but with the argument order of some predicates modified. The documentation for *rbtrees* is available at http://www.swi-prolog.org/pldoc/doc/_SWI_/library/rbtrees.pl. If we let *rbtree*(*A*,*B*) denote the type of trees mapping keys of type *A* to values of type *B*, then the types (Schrijvers et al., 2008) and meanings of the predicates used in this report are as follows:

rb_empty(*-rbtree*(*A*,*B*))

rb_empty(*T*) means *T* is an empty red-black tree.

rb_add(*+A*, *B*, *+rbtree*(*A*,*B*), *-rbtree*(*A*,*B*))

rb_add(*K*,*V*,*T*₁,*T*₂) means *K* is not in *T*₁ and *T*₂ is the result of adding value *V* under key *K* to *T*₁.

rb_in(*A*, *B*, *+rbtree*(*A*,*B*))

rb_in(*K*,*V*,*T*) means *T* contains value *V* under key *K*.

rb_app(*+pred*(*B*,*B*), *+A*, *+rbtree*(*A*,*B*), *-rbtree*(*A*,*B*))

rb_app(*P*,*K*,*T*₁,*T*₂) means *T*₂ is the same as *T*₁ except that the value associated with *K* is *V*₁ in *T*₁ and *V*₂, where *call*(*P*,*V*₁,*V*₂) is true.

rb_upd(*+A*, *B*, *B*, *+rbtree*(*A*,*B*), *-rbtree*(*A*,*B*))

rb_app(*K*,*V*₁,*V*₂,*T*₁,*T*₂) means *T*₂ is the same as *T*₁ except that the value associated with *K* is *V*₁ in *T*₁ and *V*₂ in *T*₂.

rb_fold(*+pred*(*pair*(*A*,*B*),*S*,*S*), *+rbtree*(*A*,*B*), *S*, *S*)

rb_fold(*P*,*T*,*S*₁,*S*₂) means *S*₂ is the result of folding *P* over a list of key-value pairs *K*–*V* obtained by traversing the tree.

The *store* module, also included in *genutils*, uses *rbtrees* to manage an associative map where a fresh key is allocated each time a new entry is added. The types and meanings of the main predicates are:

store_new(*-store*)

store_new(*S*) means *S* is an empty store

store_add(*A*, *-ref*(*A*), *+store*, *-store*)

store_add(*V*,*R*,*S*₁,*S*₂) means *S*₂ is the result of adding value *V* to *S*₁, and *R* is a new reference to the location in *S*₂ where *V* is stored.

store_app(*+pred*(*A*,*A*), *+ref*(*A*), *+store*, *-store*)

store_app(*P*,*R*,*S*₁,*S*₂) means *store_upd*(*R*,*V*₁,*V*₂,*S*₁,*S*₂), *call*(*P*,*V*₁,*V*₂).

store_upd(*+ref*(*A*), *A*, *A*, *+store*, *-store*)

store_upd(*R*,*V*₁,*V*₂,*S*₁,*S*₂) means *S*₁ and *S*₂ differ only in the value associated with reference *R*, which is *V*₁ in *S*₁ and *V*₂ in *S*₂.

The *tries* library module included with SWI Prolog manages a associative map from arbitrary terms as keys to mutable storage cells. Key terms which contain variables are considered to match any term in the same variant class. Value terms are copied both on insertion and retrieval. Documentation is available at <http://www.swi-prolog.org/pldoc/man?section=trie>.

References

- S. Abdallah. Memoisation: Purely, left-recursively, and with (continuation passing) style. *arXiv preprint arXiv:1707.04724*, 2017.
- W. Chen and D. S. Warren. Towards effective evaluation of general logic programs. In *in the Proceedings of the 12th PODS*. Citeseer, 1993.
- O. Danvy and A. Filinski. Abstracting control. In *Proceedings of the 1990 ACM conference on LISP and functional programming*, pages 151–160. ACM, 1990.
- P. C. de Guzman, M. Carro, and M. V. Hermenegildo. A program transformation for continuation call-based tabled execution. In M. C. L. nares and B. Demoen, editors, *Proceedings of 8th International Colloquium on Implementation of Constraint and Logic Programming Systems (CICLOPS 2008)*, pages 75–90, Udine, December 2008.
- B. Desouter, M. Van Dooren, and T. Schrijvers. Tabling as a library with delimited control. *Theory and Practice of Logic Programming*, 15(4-5):419–433, 2015.
- R. K. Dyvbig, S. P. Jones, and A. Sabry. A monadic framework for delimited continuations. Technical Report 615, Computer Science Department Indiana University, 2005.
- J. Earley. An efficient context-free parsing algorithm. *Communications of the ACM*, 13(2):94–102, 1970.
- M. Felleisen. The theory and practice of first-class prompts. In *Proceedings of the 15th ACM SIGPLAN-SIGACT symposium on Principles of programming languages*, pages 180–190. ACM, 1988.
- A. Filinski. Representing layered monads. In *Proceedings of the 26th ACM SIGPLAN-SIGACT symposium on Principles of programming languages*, pages 175–188. ACM, 1999.
- M. Johnson. Memoization in top-down parsing. *Computational Linguistics*, 21(3):405–417, 1995.
- O. Kiselyov. Delimited control in OCaml, abstractly and concretely. *Theoretical Computer Science*, 435:56–76, 2012.
- R. Leermakers. *The Functional Treatment of Parsing*. International Series in Engineering and Computer Science. Kluwer Academic Publishers, 1993.
- S. Liang, P. Hudak, and M. Jones. Monad transformers and modular interpreters. In *Proceedings of the 22nd ACM SIGPLAN-SIGACT symposium on Principles of programming languages*, pages 333–343. ACM, 1995.
- T. Lindgren. A continuation-passing style for Prolog. In *SLP*, pages 603–617. Citeseer, 1994.
- D. Michie. Memo functions and machine learning. *Nature*, 218(5136):19–22, 1968.
- U. Neumerkel. Continuation Prolog: A new intermediary language for WAM and BinWAM code generation. In *Post-ILPS'95 Workshop on Implementation of Logic Programming Languages. F16G*, 1995.
- U. Neumerkel. Lambdas und Schleifen in monotonen Logikprogrammen. In *Kolloquium Programmiersprachen und Grundlagen der Programmierung*, 2009.

- P. Norvig. Techniques for automatic memoization with applications to context-free parsing. *Computational Linguistics*, 17(1):91–98, 1991.
- F. C. Pereira and D. H. Warren. Definite clause grammars for language analysis—a survey of the formalism and a comparison with augmented transition networks. *Artificial intelligence*, 13(3):231–278, 1980.
- F. C. Pereira and D. H. Warren. Parsing as deduction. In *Proceedings of the 21st annual meeting on Association for Computational Linguistics*, pages 137–144. Association for Computational Linguistics, 1983.
- H. H. Porter. Earley deduction. Technical report, Oregon Graduate Center, 1986.
- T. Schrijvers, V. S. Costa, J. Wielemaker, and B. Demoen. Towards typed prolog. In *International Conference on Logic Programming*, pages 693–697. Springer, 2008.
- T. Schrijvers, B. Demoen, B. Desouter, and J. Wielemaker. Delimited continuations for Prolog. *Theory and Practice of Logic Programming*, 2013.
- C.-c. Shan. Shift to control. In *Proceedings of the 5th workshop on Scheme and Functional Programming*, pages 99–107, 2004.
- Y.-D. Shen, L.-Y. Yuan, J.-H. You, and N.-F. Zhou. Linear tabulated resolution based on prolog control strategy. *Theory and Practice of Logic Programming*, 1(01):71–103, 2001.
- L. Sterling and E. Y. Shapiro. *The Art of Prolog: Advanced Programming Techniques*. MIT press, 1994.
- G. J. Sussman and G. L. Steele. Scheme: A interpreter for extended lambda calculus. AI Memo 349, 1975.
- H. Tamaki and T. Sato. OLD resolution with tabulation. In *Third international conference on logic programming*, pages 84–98. Springer, 1986.
- P. Tarau and V. Dahl. Logic programming and logic grammars with binarization and first-order continuations. In *LOPSTR’94: 4th Int. Workshop on Logic Program Synthesis and Transformation, Pisa*, volume 883, 1994a.
- P. Tarau and V. Dahl. Logic programming and logic grammars with first-order continuations. In *Logic Program Synthesis and Transformation—Meta-Programming in Logic*, pages 215–230. Springer, 1994b.
- P. Tarau and V. Dahl. Mobile threads through first order continuations. In *Proceedings of APPAI-GULP-PRODE’98*, Coruna, Spain, July 1998.
- D. H. Warren. Earley deduction. *Unpublished note*, 1975.
- J. Wielemaker, T. Schrijvers, M. Triska, and T. Lager. SWI-Prolog. *Theory and Practice of Logic Programming*, 12(1-2):67–96, 2012. ISSN 1471-0684.